

Aide-mémoire minimal pour Perl/Tk

Un exemple simple d'application Perl/Tk :

```
use strict;
use Tk;
my $mainWin = new MainWindow ( -title => 'Hello' );
$mainWin->Label( -text => 'Hello, ⊔world!' )->pack();
$mainWin->Button(
    -text => 'Cliquer ⊔ici',
    -command => sub { $mainWin->Label( -text=>'Bonus' )->pack(); }
)->pack();
MainLoop();
```

La commande `new MainWindow` crée la fenêtre principale. Les autres éléments sont rattachés à cet objet.

Pour afficher un block dans son conteneur : `$block->pack()`;

Le programme se termine par `MainLoop()`; qui lance l'interface graphique. Les actions seront alors effectuées par *callback*, comme pour ce `Button`.

1 Tk::Label

Block de texte dans un widget.

`-text` Texte affiché. Exemple : `$mainWin->Label(-text => "Mon ⊔texte");`

`-textvariable` Référence à une variable texte évolutive. `$mainWin->Label(-textvariable=>\$label);`

2 Tk::Button

Bouton poussoir.

`-text` Texte du bouton. On peut aussi utiliser `-textvariable` comme ci-dessus.

`-command` Procédure lancée si le bouton est actionné.

```
$mainWin->Button( -command => \&ma_fonction );
```

Pour la syntaxe complète, voir `perldoc Tk::callback`.

3 Tk::Entry

Zone de saisie sur une ligne.

`-textvariable` Associe le contenu à une référence à un scalaire texte automatiquement mis à jour en cas de changement.

```
my $entry = $mainWin->Entry( -textvariable => \$saisie );
```

`insert()` Méthode insérant du texte à la position donnée.

```
$entry->insert(0,'Au ⊔début ⊔: ⊔');
```

`get()` Méthode pour lire le contenu actuel. `$entry->get();`

4 Tk::getOpenFile & Tk::getSaveFile

Ouvre une fenêtre de sélection de fichier et renvoie le nom du fichier choisi. Voir aussi `chooseDirectory()`.

```
$filename_label = $mainWin->getOpenFile();
```

5 Placement des éléments

Par défaut, `pack()` empile verticalement les objets. On peut changer la direction d'empilement avec `-side`, mais aussi utiliser des *frames* pour regrouper les éléments. Pour que les éléments puissent s'agrandir, utiliser `-fill` (et aussi `-expand` à l'intérieur d'une frame).

```
$mainWin->Label( -text => "Mon ⊔texte" )->pack( -side => "bottom" );
my $blockTop = $mainWin->Frame()->pack( -side => "top", -fill => "x" );
$blockTop->Label( -text => "Fichier ⊔: ⊔" )->pack( -side => "left" );
$blockTop->Entry()->pack( -side => "left", -fill => "x", -expand => 1 );
```

6 Tk::Dialog

Ouvre une fenêtre de dialogue simple. Voir Tk::DialogBox pour des dialogues plus complexes.

```
use Tk::Dialog;
# ... code partiel ...
$dialogue = $mainWin->Dialog (
    -title    => 'Confirmer' ,          # titre de la fenêtre
    -text     => 'Êtes-vous sûr?' ,    # texte principal
    -bitmap   => 'question' ,          # icone 'error', 'info', 'warning'...
    -default_button => 'Oui' ,
    -buttons  => [ 'Oui' , 'Non' ]
) ;
$reponse = $dialogue->Show(); # 'Oui' ou 'Non'
```

Pour un simple message, on peut utiliser `messageBox()`.

```
$reponse = $mainWin->messageBox(
    -title    => 'Confirmer' ,          # titre de la fenêtre
    -message  => 'Êtes-vous sûr?' ,    # texte principal
    -type     => 'YesNo' ,             # Cf Tk::messageBox
    -default  => 'Yes' ,
);
```

7 Tk::Menu

Afficher des menus déroulants. On peut les placer dans une barre de menu.

```
use Tk ;
$mainWin = new MainWindow();
$barre_menu = $mainWin->Frame(
    -relief => 'groove' , -borderwidth => 2
);
$menu_fichier = $barre_menu->Menubutton (
    -text => 'Fichier' ,
    -tearoff => 0 ,          # non déplaçable
    -menuitems => [
        [ 'command' => 'Ouvrir' ] ,
        '-' ,
        [ 'command' => 'Quitter', -command => \&quitter ]
    ]
)->pack( -side => 'left' );

$barre_menu->pack( -side => 'left', -anchor => 'n' );
MainLoop ( ) ;
```

8 Pour aller plus loin

- La doc officielle : `perldoc Tk::UserGuide` (aussi disponible en ligne sur CPAN.org)
- L'application Perl/Tk de démonstration (avec le code source dans l'interface) : `widgit`
- Un article des mongueurs de Perl : <http://articles.mongueurs.net/magazines/linuxmag62.html>
- *Perl/Tk - précis & concis*, chez O'Reilly, 112 pages.

Copyright (c) 2007-2009 François Gannaz (francois.gannaz@silecs.info)

Permission vous est donnée de copier, distribuer et/ou modifier ce document selon les termes de la Licence GNU Free Documentation License, Version 2.0 ou ultérieure publiée par la Free Software Foundation; pas de section inaltérable; pas de texte inaltérable de première page de couverture; texte inaltérable de dernière page de couverture : « Auteur : François Gannaz (francois.gannaz@silecs.info) »